abc.nb on July 26, 2015

July-26-15 2:54 AM

Pensieve header: One-Co computations in the abc presentation.

The bracket

On the elements β, a, c, δa, ca, δaa.

Generalities

```
DQ[is___] := (Sort[{is}] === Union[{is}]);
OQ[is___] := OrderedQ[{is}];

Simp[expr_] := Simplify[expr];
S[β[f_]] := β[Simp[f]];
S[a[i_, j_]] := a[i, j];
S[a[f_, i_, j_]] := a[Simp[f], i, j];
S[c[f_, k_]] := c[Simp[f], k];
S[δa[f_, i_, j_]] := δa[Simp[f], i, j];
S[ca[f_, j_, k_, l_]] := ca[Simp[f], j, k, l];
S[δaa[f_, i_, j_, k_, l_]] := δaa[Simp[f], i, j, k, l];
S[expr_] := expr /. (λ_β | λ_a | λ_δa | λ_c | λ_ca | λ_δaa) :> S[λ];

β[0] := 0;
β /: β[f_] + β[g_] := β[f + g];
β /: g_ * β[f_] := β[g f] // S;
a[0, _, _] := 0;
a /: a[f_, j_, k_] + a[g_, j_, k_] := a[f + g, j, k];
a /: g_ * a[f_, j_, k_] := a[g f, j, k] // S;
c[0, _] := 0;
c /: c[f_, j_] + c[g_, j_] := c[f + g, j];
c /: g_ * c[f_, j_] := c[g f, j] // S;
δa[0, _, _] := 0;
δa /: δa[f_, j_, k_] + δa[g_, j_, k_] := δa[f + g, j, k];
δa /: g_ * δa[f_, j_, k_] := δa[g f, j, k] // S;
ca[0, _, _, _] := 0;
ca /: ca[f_, j_, k_, l_] + ca[g_, j_, k_, l_] := ca[f + g, j, k, l];
ca /: g_ * ca[f_, j_, k_, l_] := ca[g f, j, k, l] // S;
δaa[0, _, _, _, _] := 0;
δaa /: δaa[f_, i_, j_, k_, l_] + δaa[g_, i_, j_, k_, l_] := δaa[f + g, i, j, k, l];
δaa /: g_ * δaa[f_, i_, j_, k_, l_] := δaa[g f, i, j, k, l] // S;
```

Can these be unified?

δaa relations

Locality:

```
S[δaa[f_, i_, j_, k_, l_]] /; ({i, j} ⋂ {k, l} === {}) ∧ ! OQ[i, k] :=
  δaa[f, k, l, i, j] // S;
```

Standard Swinging:

∧ OQ[i,k]

```
S[δaa[f_, i_, j_, k_, l_]] /; DQ[i, j] ∧ DQ[k, l] ∧ ! OQ[j, l] := S[Expand[
    δaa[f, i, l, k, j] +
     ϵ_1 (ca[b_k f, l, i, j] - ca[b_i f, l, k, j] - ca[b_k f, j, i, l] + ca[b_i f, j, k, l])
   ]];
```

1322 Swinging:

```
S[δaa[f_, i_, j_, k_, k_]] /; DQ[i, j, k] ∧ OQ[i, k, j] := S[Expand[
    δaa[f, i, k, k, j] +
     ϵ_1 (ca[b_k f, k, i, j] - ca[b_i f, k, k, j] - ca[b_k f, j, i, k] + ca[b_i f, j, k, k]) +
     ϵ_3 (δa[b_i f, k, j] - c[b_i b_k, j])
   ]];
```

Commute Heads:

```
S[δaa[f_, i_, k_, j_, k_]] /; DQ[i, j, k] ∧ ! OQ[i, j] := S[Expand[
    δaa[f, j, k, i, k] + ϵ_2 (δa[b_i f, j, k] - δa[b_j f, i, k])
   ]];
```

NonCommutativeMultiply

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[0, _] = 0; NonCommutativeMultiply[_, 0] = 0;
NonCommutativeMultiply[x_, x_] = 0;
NonCommutativeMultiply[x_Plus, y_] := NonCommutativeMultiply[#, y] & /@ x;
NonCommutativeMultiply[x_, y_Plus] := NonCommutativeMultiply[x, #] & /@ y;
```

```
β[f_] ** a[g_, j_, k_] := a[f g, j, k];
β[f_] ** c[g_, j_] := c[f g, j];
c[g_, j_] ** β[f_] := c[f g, j];
β[f_] ** δa[g_, j_, k_] := δa[f g, j, k];
β[f_] ** ca[g_, i_, j_, k_] := ca[f g, i, j, k];
β[f_] ** δaa[g_, i_, j_, k_, l_] := δaa[f g, i, j, k, l];
δa[g_, j_, k_] ** β[f_] := δa[f g, j, k];
δ ** a[f_, i_, j_] := δa[f, i, j];
c[f_, i_] ** a[g_, j_, k_] := ca[f g, i, j, k];
a[f_, i_, j_] ** δa[g_, k_, l_] := δaa[f g, i, j, k, l];
δa[f_, i_, j_] ** a[g_, k_, l_] := δaa[f g, i, j, k, l];
```

```
δ ** _ca = 0;
δ ** _δaa = 0;
_c ** _ca = _ca ** _c = 0;
_c ** _δaa = _δaa ** _c = 0;
_δa ** _δaa = _δaa ** _δa = 0;
_δa ** _ca = _ca ** _δa = 0;
```

```
NonCommutativeMultiply::ndef =
 "NonCommuatativeMultiply is not defined on {`1`,`2`}."
NonCommutativeMultiply[x_, y_] :=
  (Message[NonCommutativeMultiply::ndef, x, y]; Undefined);
```

```
NonCommuatativeMultiply is not defined on {`1`,`2`}.
```

Bracket Generalities

```
B[0, _] = 0; B[_, 0] = 0;
B[x_, x_] = 0;
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
```

The γ shortcuts

```
γ[f_, j_, k_] /; DQ[j, k] := δa[f, j, k] - c[b_j f, k] // S;
γ[f_, j_, k_, l_] /; DQ[j, k, l] := ca[f, l, j, k] - ca[f, k, j, l] // S;
```

Fundamental Brackets

a-β, a-c, a-a, AS

```
B[a[j_, k_], β[g_]] := γ[∂_{b_j} g - ∂_{b_k} g, j, k];
B[β[g_], a[j_, k_]] := -B[a[j, k], β[g]];
B[a[j_, k_], a[l_, m_]] /; DQ[j, k, l, m] := 0;
B[a[j_, k_], a[j_, l_]] /; DQ[j, k, l] := γ[1, j, k, l] // S;
B[a[j_, k_], a[i_, k_]] /; DQ[i, j, k] := a[b_i, j, k] - a[b_j, i, k] // S;
B[a[j_, k_], a[k_, l_]] /; DQ[j, k, l] := a[b_j, k, l] - a[b_k, j, l] - γ[1, j, k, l] // S;
B[a[k_, l_], a[j_, k_]] /; DQ[j, k, l] := -B[a[j, k], a[k, l]];
B[a[f_, j_, k_], c[g_, j_]] /; DQ[j, k] := γ[-f g, j, k];
B[a[f_, j_, k_], c[g_, k_]] /; DQ[j, k] := γ[f g, j, k];
B[a[f_, j_, k_], c[g_, l_]] /; DQ[j, k, l] := 0;
B[c[g_, l_], a[f_, j_, k_]] := -B[a[f, j, k], c[g, l]];
```

Vanishing brackets

```
B[_β, _β | δ | _c | _δa | _ca | _δaa] = 0;
B[_β | δ | _c | _δa | _ca | _δaa, _β] = 0;
B[δ | _c | _δa | _ca | _δaa, δ | _c | _δa | _ca | _δaa] = 0;
```

Composite Brackets

```
B[a[f_, j_, k_], β[g_]] := β[f] ** B[a[j, k], β[g]];
B[β[g_], a[f_, j_, k_]] := -B[a[f, j, k], β[g]];
B[a[f_, j_, k_], a[l_, m_]] :=
  B[β[f], a[l, m]] ** a[l, j, k] + β[f] ** B[a[j, k], a[l, m]];
B[a[f_, j_, k_], a[g_, l_, m_]] :=
  B[a[f, j, k], β[g]] ** a[l, l, m] + β[g] ** B[a[f, j, k], a[l, m]];
B[a[f_, i_, j_], δa[g_, k_, l_]] := δ ** B[a[f, i, j], a[g, k, l]];
B[δa[f_, i_, j_], a[g_, k_, l_]] := δ ** B[a[f, i, j], a[g, k, l]];
B[a[f_, i_, j_], ca[g_, k_, l_, m_]] :=
  B[a[f, i, j], c[g, k]] ** a[l, l, m] + c[g, k] ** B[a[f, i, j], a[l, m]];
B[ca[g_, k_, l_, m_], a[f_, i_, j_]] := -B[a[f, i, j], ca[g, k, l, m]];
B[a[f_, i_, j_], δaa[g_, k_, l_, m_, n_]] :=
  B[a[f, i, j], δa[g, k, l]] ** a[1, m, n] + δa[g, k, l] ** B[a[f, i, j], a[m, n]];
B[δaa[g_, k_, l_, m_, n_], a[f_, i_, j_]] := -B[a[f, i, j], δaa[g, k, l, m, n]];

B::ndef = "B is not defined on {`1`,`2`}."
B[x_, y_] := (Message[B::ndef, x, y]; Undefined);

B is not defined on {`1`,`2`}.
```

Testing Jacobi and Anti-Symmetry

```
FormalPlusBasis[n_, f_] := Module[{ff},
   ff = f @@ Table[b_i, {i, n}];
   Flatten@{
      β[ff],
      Table[a[ff, i, j], {i, n - 1}, {j, i + 1, n}],
      Table[c[ff, i], {i, n}],
      Table[δa[ff, i, j], {i, n - 1}, {j, i + 1, n}],
      Table[ca[ff, i, j, k], {i, n}, {j, n - 1}, {k, j + 1, n}],
      Table[δaa[ff, i, j, k, l], {i, n - 1}, {j, i + 1, n}, {k, n - 1}, {l, k + 1, n}]
     } /. 1[___] → 1
  ];
```

```
FormalPlusBasis[3, f]
```

```
{β[f[b1, b2, b3]], a[f[b1, b2, b3], 1, 2], a[f[b1, b2, b3], 1, 3], a[f[b1, b2, b3], 2, 3],
 c[f[b1, b2, b3], 1], c[f[b1, b2, b3], 2], c[f[b1, b2, b3], 3], δa[f[b1, b2, b3], 1, 2],
 δa[f[b1, b2, b3], 1, 3], δa[f[b1, b2, b3], 2, 3], ca[f[b1, b2, b3], 1, 1, 2],
 ca[f[b1, b2, b3], 1, 1, 3], ca[f[b1, b2, b3], 1, 2, 3], ca[f[b1, b2, b3], 2, 1, 2],
 ca[f[b1, b2, b3], 2, 1, 3], ca[f[b1, b2, b3], 2, 2, 3], ca[f[b1, b2, b3], 3, 1, 2],
 ca[f[b1, b2, b3], 3, 1, 3], ca[f[b1, b2, b3], 3, 2, 3], δaa[f[b1, b2, b3], 1, 2, 1, 2],
 δaa[f[b1, b2, b3], 1, 2, 1, 3], δaa[f[b1, b2, b3], 1, 2, 2, 3],
 δaa[f[b1, b2, b3], 1, 3, 1, 2], δaa[f[b1, b2, b3], 1, 3, 1, 3],
 δaa[f[b1, b2, b3], 1, 3, 2, 3], δaa[f[b1, b2, b3], 2, 3, 1, 2],
 δaa[f[b1, b2, b3], 2, 3, 1, 3], δaa[f[b1, b2, b3], 2, 3, 2, 3]}
```

```
AS[x1_, x2_] := Module[{as},
   as = B[x1, x2] + B[x2, x1] // S;
   If[as === 0, as, {x1, x2} → as]
  ];
DeleteCases[Flatten[Outer[
   AS,
   FormalPlusBasis[3, f],
   FormalPlusBasis[3, g]
  ]], 0]
```

```
{}
```

```
Jacobi[x1_, x2_, x3_] := Module[{Jac},
   Jac = S[B[x1, B[x2, x3]] + B[x2, B[x3, x1]] + B[x3, B[x1, x2]]];
   If[Jac === 0, Jac, {x1, x2, x3} → Jac]
  ];
```

```
Jacobi @@ {a[f, 1, 2], a[g, 1, 3], a[h, 2, 3]} // S
```

```
{a[f, 1, 2], a[g, 1, 3], a[h, 2, 3]} → ca[-f g h b1, 2, 2, 3] + ca[f g h b1, 3, 2, 2] +
  ca[-f g h b2, 3, 1, 2] + ca[f g h b2, 2, 1, 3] + δaa[-f g h, 1, 2, 2, 3] +
  δaa[-f g h, 1, 3, 2, 3] + δaa[f g h, 2, 2, 1, 3] + δaa[f g h, 2, 3, 1, 3]
```

```
JacErrors = DeleteCases[
  bas1 = FormalPlusBasis[4, f];
  bas2 = FormalPlusBasis[4, g];
  bas3 = FormalPlusBasis[4, h];
  Flatten[
   Table[Jacobi[bas1⟦i⟧, bas2⟦j⟧, bas3⟦k⟧],
    {i, Length[bas1] - 1}, {j, i + 1, Length@bas2}, {k, i + 1, Length@bas3}]
  ],
  0]
```

B::ndef : B is not defined on {a[2, 3],a[3, 2]}.

NonCommutativeMultiply::ndef : NonCommuatativeMultiply is not defined on {β[g[b_1, b_2, b_3, b_4]],Undefined}. ≫

NonCommutativeMultiply::ndef : NonCommuatativeMultiply is not defined on {c[f[b_1, b_2, b_3, b_4] h[b_1, b_2, b_3, b_4] b_1^2, 4],Undefined}. ≫

B::ndef : B is not defined on {a[2, 3],a[3, 2]}.

NonCommutativeMultiply::ndef : NonCommuatativeMultiply is not defined on {β[g[b_1, b_2, b_3, b_4]],Undefined}. ≫

General::stop : Further output of NonCommutativeMultiply::ndef will be suppressed during this calculation. ≫

B::ndef : B is not defined on {a[2, 3],a[3, 2]}.

General::stop : Further output of B::ndef will be suppressed during this calculation. ≫

{ ⋯ 1 ⋯ }

large output | show less | show more | show all | set size limit...

```
JacErrors // Length
```

221